NORMAL SUBGROUPS IN MAPLE

Link to: physicspages home page.

To leave a comment or report an error, please use the auxiliary blog and include the title or URL of this post in your comment.

Post date: 16 January 2026.

Reference: *A Gentle Introduction to Group Theory*, Bana Al Subaiei & Muneerah Al Nuwairan, Section 7.5.

Maple can be used to generate the normal subgroups of a group G. It can also be used to test if a subgroup is a normal subgroup. For example, using the symmetric group $\mathfrak{S}_3$ for the permutation of 3 objects, we can write

```
with(GroupTheory);
S3 := Symm(3);
S3ns := NormalSubgroups(S3);
```

This produces the list

```
S3ns := NormalSubgroups(S3)
                S3ns := [<(23)(12)>, <(123)>, <>]
```

Each subgroup is given in generator notation. To see the actual elements in each subgroup, we can use `Elements`. For example, the subgroup $\langle(123)\rangle$ has the elements:

```
Elements(S3ns[2]);
            {(),(132),(123)}
```

To test if a group is a normal subgroup, we can use `IsNormal`. For example, if we generate all subgroups of $\mathfrak{S}_3$ and then test one of them, we have

```
S3s := SubgroupLattice(S3);
S3sl := convert(S3s, 'list');
```

Element 2 of `S3sl` is the subgroup $\langle(2,3)\rangle = \{(),(2,3)\}$. We can then test if it's normal:

```
IsNormal(S3sl[2],S3);
              false
```

If we want to examine the actual calculations for determining if a subgroup H is normal, we can use the results of an earlier post. In particular, we can test if $aha^{-1} \in H$, where h is an element of H and a is an element

of the original group G. The following Maple code will run through the calculations for a given group `G` and one of its subgroups `H`.

```
ns := proc(G, H)
 local Hel, Gel, i, a, am1, j, h, aham1, ismem;
 Hel := Elements(H);
 Gel := Elements(G);
 print("subgroup: ", Hel);
 for i from 1 to numelems(Gel) do
  a := Gel[i];
  am1 := a^(-1);   #Inverse of a
  for j from 1 to numelems(Hel) do
   h := Hel[j];
   aham1 := (a . h) . am1;
   ismem := member(aham1, Hel);
   print(a, h, am1, aham1, ismem);
   end do;
  end do;
  print("Num checks = ", numelems(Hel)*numelems(Gel));
end proc
```

The first print statement prints out the elements of the subgroup `H`. The line `print(a, h, am1, aham1, ismem);` prints out the elements a, h, a^{-1}, aha^{-1} followed by a boolean quantity `ismem` which tests if $aha^{-1} \in H$. If `ismem` is true for all tests, then the subgroup is normal, otherwise it isn't. The "Num checks" printed out at the end gives the total number of checks that were done. Running this procedure on the group $\mathfrak{S}_3$ and the subgroup $\langle(2,3)\rangle$ we find:

```
ns(S3, S3sl[2]);
                    "subgroup: ", {(), (23)}
                      (), (), (), (), true
                     (), (23), (), (23), true
                     (13), (), (13), (), true
                  (13), (23), (13), (12), false
                     (23), (), (23), (), true
                   (23), (23), (23), (23), true
                    (132), (), (123), (), true
                 (132), (23), (123), (13), false
                    (123), (), (132), (), true
                 (123), (23), (132), (12), false
```

```
(12), (), (12), (), true
(12), (23), (12), (13), false
"Num checks = ", 12
```

Since some of the checks return 'false' this subgroup is not normal.

Running the procedure on `S3sl[5]`, we have

```
ns(S3, S3sl[5]);
"subgroup: ", {(), (132), (123)}
(), (), (), (), true
(), (132), (), (132), true
(), (123), (), (123), true
(13), (), (13), (), true
(13), (132), (13), (123), true
(13), (123), (13), (132), true
(23), (), (23), (), true
(23), (132), (23), (123), true
(23), (123), (23), (132), true
(132), (), (123), (), true
(132), (132), (123), (132), true
(132), (123), (123), (123), true
(123), (), (132), (), true
(123), (132), (132), (132), true
(123), (123), (132), (123), true
(12), (), (12), (), true
(12), (132), (12), (123), true
(12), (123), (12), (132), true
"Num checks = ", 18
```

In this case, all checks return 'true' so this is a normal subgroup.

Running the procedure on groups much larger than this is a bit impractical, as it produces far too many checks to be useful.